

JMPer Cable

Special Scripting Edition

Inside This Newsletter

Articles in this special issue of the JMPer Cable focus on JMP Scripting Language (JSL) and its application in a variety of fields and circumstances.

Dynamic Illustrations in the Classroom p. 5

ANOM in JMP p. 7

Creating a List of Open Data Tables p. 10

Meet the Trainer: An Interview p. 11

Breaking News

Join us for the second annual JMP User Conference on June 7-8, 2005 at the SAS World Headquarters in Cary, North Carolina.

The conference brings together users of all levels and includes sessions for JMP novices. It will be followed by two days of training on June 9-10.

For details, see <http://support.sas.com/training/jmp/index.html>

Using SAS IOM Commands in JSL

Matt Flynn, SAS Institute

The SAS IOM commands available in JMP Scripting Language (JSL) make it possible to connect directly from within JMP to a SAS server to retrieve data for further analysis within JMP.

The script presented in this article focuses on library submission, highlighting the SAS IOM commands for connecting and executing SAS jobs, and returning the results back to JMP for further processing. For a complete, downloadable version of the script, with comments, see `Connect_to_SAS.jsl` at <http://www.jmp.com/news/jmpercable>.

Background

I had positive experiences connecting other PC desktop tools to remote SAS servers via SAS Integration Technologies (SAS/IT) and saw that with the addition of the IOM commands, it would be possible to achieve this using JSL. Automating the often laborious multi-step process of retrieving data would allow ease of use in performing statistical analysis in JMP.

About the Script

This script allows a Windows JMP environment to talk to a remote SAS server. It connects to external servers and retrieves data in a point-and-click

environment. The script enables an automated method of launching (possibly remote) SAS batch jobs, retrieving SAS datasets (which may be the results of those batch jobs or simply the latest update of required transactional data) and imports the data seamlessly into JMP for statistical analysis and reporting. It eliminates the tedious, manual process of connecting to servers, navigating to the data, retrieving the data (perhaps via FTP), bringing that data into JMP, and then performing an analysis. The script allows the analysis to receive full attention and concentration. Even better, the script can be shared with JMP users who are not JSL programmers, enabling them to easily retrieve the remote data they need.

The script presented in this article is specific to a machine that has access to a local copy of PC SAS or a remote UNIX server running SAS. The following example sets up the script to connect to the two SAS platforms, execute SAS batch jobs, retrieve data (possibly the results of those batch jobs), and pulls that data into JMP for either interactive analysis or to repeat pre-programmed analysis platforms.

Connecting to a SAS Server

The first step is to write the part of the script that asks the user to select between a local and a UNIX server:

```
result = Dialog( title( "Run a SAS batch job" ) ,
    vlist(
        lineup( 2 ,
            "Select Server" , server = ComboBox( "local" , "UNIX" ) ,
            hlist( Button( "OK" ) , Button( "Cancel" ) )
        )
    )
);
```

This creates the window in Figure 1.




Figure 1: Ask the user to choose from available servers.

Now we want JMP to parse the results of the dialog box into variables. The following JSL does this:

```
show( result ) ;
server = result[ "server" ] ; show( server ) ;
button = result[ "Button" ] ; show( button ) ;
```

In this script, server 1 is the local server and server 2 is the remote server.

Furthermore, button == 1 means that the user pressed OK and button == -1 means that the user pressed Cancel. The JSL IOM command `sas connect("servername" , portID)` calls SAS on either the local machine or a remote UNIX host. Insert the appropriate name and port ID of your target SAS server here. The resulting dialog prompts for the appropriate userid and password:

```
if ( button == 1 ,
    // display user selections in the log
    if ( server == 1 ,
        show( "local", server ) ,
        show( "UNIX", server )
    );
    // connect to local or remote SAS instance
    if ( server == 1 ,
        sas connect( "localhost" , 8591 ) ,
        sas connect( "unix.sas.com" , 5310 )
    );
);
```

This creates the window in Figure 2.




Figure 2: Checking User ID and password.

For verbal feedback, you could display a verbal and visual message to the user while

the SAS server starts up:

```
caption( { 400 , 400 } , spoken(
    1 ) , Delayed( 0 ) ,
    "Connecting to SAS
    server..." );
wait( 3 );
caption( remove );
```

Retrieving SAS Files

Next, the script retrieves a list of files in a certain directory. In this script the directory is hard-coded to look at specific directories, both local and on the remote server. You should assign the desired directory name for your needs. For example, the script looks at the local directory C:\temp and the remote UNIX home directory ~/temp. You could generalize the script to branch based upon User ID, for example:

```
if ( server == 1 ,
    path = "C:\temp" ,
    path = "~/temp/"
);
show( "path" , path );

if ( server == 1 ,
    filelist = sas get file
    names in path( path ) ,
    filelist = sas get
    file names in
    path( "~/temp/" );
);
show( filelist );
```

Next, the script queries the directory and retrieves a list of files located in that directory. At this point, the list is limited to only the executable SAS program files available in this directory, so that the user can choose one to execute.

```
sasFiles = { } ;
for ( i = 1, i <= n items(
    fileList ) , i++ ,
    len = length( fileList[ i ] )
);
if ( len > 4 ,
    if ( lowercase( substr(
        fileList[ i ] , len -
        3 ,
        len ) ) == ".sas" ,
```

```

        insertinto( sasFiles,
                    fileList[ i ] ) ) ;
    ) ;
    ) ;
show( sasFiles ) ;

```

Executing a SAS Job

Now display a dialog prompting the user to select a SAS program to run:

```

dlg = Dialog( "Select SAS program
             to run",
             vlist(
                 selection = listBox(
                     sasFiles ) ,
                 hlist( Button("OK"),
                       Button( "Cancel"
                               ) ) )
             ) ;
sel = dlg[ "selection" ] ; show(
sel ) ;

```

This creates the window in Figure 3.




Figure 3: Select a SAS program to run.

Optionally Asking for Parameters for the SAS Job

For selected SAS programs, prompt for a user-specified parameter. If the user selects a specified program (in this case, the dialog is hard-coded to appear for the data file x.sas), the user is prompted to input parameters to add flexibility to the SAS batch job call. This one simply asks the user how many observations to process during the job.

```

if ( sel == {"x.sas"} ,
    result = Dialog( title(
        "Please specify how many Obs to
        run" ),
        vlist(
            lineup( 2 ,
                "Set number of obs",
                obs = EditNumber( 10 ) ,

```

```

                    hlist( Button( "OK" ) , Button( "Cancel" ) )
                )
            )
        )
    );

```

This creates the window in Figure 4.




Figure 4: Optionally ask for parameters for the SAS job.

Next, the results of the dialog box are parsed into JSL variables:

```

show ( sel ) ;
if ( sel == {"x.sas"} ,
    obs = result[ "obs" ] ; show( obs )
);

```

If you're using JMP 5.1.1 and later, use the standard way to submit an existing SAS program file:

```
SAS Submit File( "C:\test\x.sas" ) ;
```

If you're using JMP 5.1 and earlier, use this workaround:

```
SAS Submit( "%include C:\test\x.sas" ) ;
```

It is a little tricky to pass the value of above selected program name and a parameter into the SAS submit string below. In this example, a SAS Macro command is built using JMP's concat function then submitted to SAS.

```

if ( sel == {"x.sas"} ,
    pgm = concat( "%let obs=", char(obs) , " ; %put *** obs = " ,
char(obs) , " ; " , " %include 'c:\temp\'", sel[1] , " ' ; run ; " ) ,
    pgm = concat( " %include 'c:\temp\'", sel[1] , " ' ; run ; " )
);
show(pgm) ;

```

Now submit the job with this command:

```
SAS Submit(pgm) ;
```

Retrieving the SAS job

Now that you have run the job, check for the success of the remote SAS job using the IOM command SAS Get Log():

```
log = SAS Get Log(); show(log);
```

The output can be retrieved into the JSL log using:

```
SAS Get Output();
```

You can specify details about opening the SAS data files. It's a good idea to first list all files in directory (some created in x.sas or y.sas), and then to trim the list to just SAS data files (*.sas7dcat).

```

filelist = sas get file names in path( path ) ; show( filelist ) ;
sasFiles = {} ;
for( i = 1 , i <= n items( filelist ) , i++ ,
    len = length( filelist[ i ] ) ;
    if ( len > 7,

```

```

        if ( lowercase( substr( fileList[ i ] , len -7 , len ) ) ==
            "sas7bdat" ,
            insertinto( sasFiles , fileList[i] )
        ) ;
    ) ;
) ;

sasFiles = sort list( sasFiles ) ;

show( sasFiles ) ;

dlg2 = Dialog( "Select dataset to open" ,
    vlist(
        lineup( 3 ,
            selection = listBox( sasFiles ) ,
            vlist( Button("OK"), Button("Cancel") )
        )
    )
) ;

sel2 = dlg2[ "selection" ] ; show( sel2 ) ;

show( path ) ;
show( server ) ;
if ( server == 1 ,
    dsn2 = concat( "C:\temp\" , sel2[1] ) ; show( dsn2 ) ; open(
        dsn2 ) ,
    dsn1 = concat( "~/temp/" , sel2[1] ) ; show( dsn1 ) ;
    dsn2 = concat( "C:\temp\" , sel2[1] ) ; show( dsn2 ) ;
    sas get file ( dsn1 , dsn2 ) ;
    open( dsn2 ) ;

) ;

show( path ) ; show( dsn2 ) ;

```

This creates the window in Figure 5.




Figure 5: Display a list of SAS datasets (some could be output files of the SAS batch job).

Running a JMP Analysis

Now, run a JMP analysis on the SAS dataset after prompting the user to select analysis columns. This example shows running a one-way analysis.

```

Oneway( Each Pair( 1 ) , All Pairs( 1 ) , Quantiles( 1 ) , Means( 1 ) ,
Means and Std Dev(1), Unequal Variances(1), Box Plots( 1 ) ,
Mean Diamonds( 1 ) , Mean Error Bars( 1 ) , Std Dev Lines( 1 ) , Comparison
Circles( 1 ) , title( "Sample Oneway Analysis" ) ) ;

```

Note: Assignment of Y and X is omitted: Y(:Height), X (:Age)




Figure 6: Resulting oneway dialog incorporating data retrieved from the remote SAS server.

Cleanup

Close the SAS session on the server when done.

```
SAS Disconnect( ) ;
```

Conclusion

JSL and the SAS IOM commands make it easy to “reach out” and retrieve data from across an organization, either as a nightly scheduled job, a query against a multi-million-row SAS dataset, or even large files in a DBMS, via SAS/ACCESS. One can easily import them into the friendly confines of JMP for either scripted or interactive analysis.

The author would like to thank JMP technical support for help creating a tricky part of the script that displays the available SAS program files so the user can choose one to execute.

Dynamic Illustrations in the Classroom

By Wayne J. Levin, JMP Instructor,
Predictum Management Sciences

For students, understanding one statistical concept generally requires understanding several statistical concepts, since many of them are inextricably linked to others. Understanding power, for example, requires an understanding of standard errors, confidence intervals, analysis of variance, hypothesis tests, and F -tests.

Instructors may now dynamically demonstrate such statistical concepts using JMP Scripting Language (JSL) so students can visualize how statistical concepts behave and how they are related. Such visualization makes comprehension faster, easier, and more readily reinforced. This article

describes the use of one JSL script and how it shows linked statistical concepts.

The Script

The script used in this article, demoOneway.jsl, can be downloaded at <http://www.jmp.com/news/impercable>. Run the script to see the window in Figure 7, which illustrates the purpose of power and how power is linked to other statistical concepts. Download a detailed description of the window below (PowerANOVACheatSheet.pdf) from <http://www.jmp.com/news/impercable>.

Click and drag the sliders to make changes to the parameters. This causes

changes in the diamonds, the p -value, the power bars, and other areas of the illustration.

During class, instructors use this window to explain what other statistics “look like” when the sample size is 10. They see that power is low (0.45818) because the confidence intervals, as depicted by the height of the diamonds, are big. In other words, the variability overwhelms the difference in averages between the two groups. The conclusion is “Fail to Reject H_0 ” with a p -value of above 0.05, as shown on the p -value chart.

When the instructor slides the sample size slider from 10 to 30 (as shown in Figure 8), the window changes to




Figure 7: Window produced by demoOneway.jsl.




Figure 8: The window changes as the instructor moves the sample size slider from 10 to 30.

show the effects. Students see the diamonds shrink (the mean diamonds became tighter) as the standard error drops, the p -value descends, and power becomes stronger, occupying almost the entire bar on the right (0.94105). As the p -value drops below alpha, the conclusion line changes to “Reject H_0 .”

Advanced Options

The script provides a set of additional reports to further clarify power. When the Show option is selected for ANOVA (Analysis of Variance), the script reveals an ANOVA report. Figure 9 shows ANOVA reports that contain a depiction of the F distribution under both the null and alternate hypotheses with a pointer to the current F -ratio.

Instructors can use the tables to compare sample sizes with all other parameters fixed. They can move the parameter sliders to update the table.

Similarly, the script also provides a power plot that adjusts in real time as the sliders are manipulated (Figure 10). Using the power plot, it is easy to see




Figure 9: ANOVA reports with N=10 (top) and N=30 (bottom).




Figure 10: Power plots with higher (left) and lower (right) standard deviation.

that when the standard deviation is high, there is little marginal gain in power from an additional observation. Reducing variability is its own reward, as seen on the right, because when the standard deviation is low, power rises considerably with each additional observation collected.

Conclusion

This script has been well received by students in my classroom. With it, students visualize concepts faster and easier—I can almost see the light bulbs going off in their minds. They are also eager to have copies of the script after the training session so they can reinforce their understanding. The script helps them apply these concepts to their analyses.

ANOM in JMP

Karen Copeland, Boulder Statistics

The Analysis of Means (ANOM) is a graphical multiple comparison procedure for comparing a group of means, rates, or proportions to the overall mean rate or proportion. The goal is to identify any group that appears different from the overall mean. The ANOM is specifically designed to compare a group of means to the overall mean. This differentiates it from ANOVA (which is more general) and the Tukey-Kramer procedure (which is specific to pairwise differences between means). One key benefit to using the ANOM over the ANOVA is the graphical component to the ANOM procedure. The ANOM decision chart, explained below, is an excellent tool for understanding the data analysis.

The ANOM decision chart is similar to a control chart in that it has a center line and decision limits. The group means are plotted against the decision limits; those that fall beyond the limits are statistically different from the overall mean. Laplace first used an ANOM-type procedure in 1827. The graphical form of the procedure and the phrase “analysis of means” were proposed by Ott (1967) and at that point in time approximate critical values were used. Exact critical values have been available since the early 80’s when they were obtained by Peter Nelson (1982). These exact critical values are based on evaluating a negatively equi-correlated k -dimensional singular t -distribution

and, thankfully, are tabulated for the rest of us to use. Tables of critical values can be found in, for example, Nelson (1982), Nelson (1983), Mason *et al.*, (1989), Vardeman (1994), and Nelson, *et al.* (2002).

ANOM assumes that the data are at least approximately normal and that the different treatments all have the same variance. For equal sample sizes, the decision limits are computed using

$$\bar{\bar{y}} \pm h(\alpha, I, N-I) \sqrt{MSE} \sqrt{\frac{I-1}{N}}$$

where

- $\bar{\bar{y}}$ = overall mean
- $h(\alpha, I, N-I)$ = tabulated critical value
- I = number of means being compared
- N = total number of data points

$$MSE = \frac{\sum_{i=1}^I s_i^2}{I}$$

For unequal sample sizes, the decision limits (see Nelson (1989) for details) are computed using:

$$\bar{\bar{y}} \pm m(\alpha, I, N-I) \sqrt{MSE} \sqrt{\frac{N-n_i}{Nn_i}}$$

where

- $\bar{\bar{y}}$ = overall mean
- $m(\alpha, I, N-I)$ = tabulated critical value
- I = number of means being compared
- N = total number of data points
- n_i = sample size of the i^{th} treatment

$$MSE = \frac{\sum_{i=1}^I (n_i - 1) s_i^2}{N - I}$$

The critical values $m(\alpha, I, N-I)$ are Studentized Maximum Modulus quantiles that are found by numerical integration and, like the h values of the equal sample size case, are tabled in references such as Nelson (1989), Nelson *et al.* (2002), and Nelson *et al.* (2005).

The studentized maximum modulus distribution is the distribution for

$$M(k, v) = \max |T_i|$$

where T_i has a univariate t distribution with v degrees of freedom.

Notice that in the unequal case, the ANOM decision chart has different decision limits for each different treatment sample size.

The ANOM procedure can also be extended to comparing rates and proportions (see Nelson *et al.* (2002) or Nelson *et al.* (2005)).

JMP Scripts for ANOM

ANOM is not a standard procedure in JMP. However, scripts generate ANOM decision charts for comparing means with either equal or unequal sample sizes.

You can download these scripts at <http://www.jmp.com/news/jimpercable>. The script for equal sample sizes is named ANOMB.jsl and the script for unequal sample sizes is named ANOMUB.jsl.

The scripts use look-up files which contain the critical values for two to eight means and significance levels

$\alpha=0.1, 0.05, 0.01, 0.001$. The look-up files can also be downloaded from <http://www.jmp.com/news/impercable>. They are Exact Factors for Balanced ANOM.jsl and Exact Factors for ANOMUB.jsl. These files must reside in the same folder as the corresponding scripts, or the statement calling the look-up file in the script needs to include the path to the look-up file.

Balanced Example: Tube Weight Data

A company that produces a medical gel filling product packaged in 4oz. (113.4 g) tubes recently purchased a new filling machine. One regulatory requirement is that they must validate the filling process. That is, they must demonstrate that the machine fills the tubes to the proper amount when using specified machine settings.

As a first step in the validation process, four machine settings were studied to understand their impact on the tube fill weights. A partial view of the data from seven tubes at each of the four settings is given in Figure 11. This data table (fweights.jmp) can be downloaded from <http://www.jmp.com/news/impercable>.

	Treatment	Weight
1	1	119.5
2	1	119.9
3	1	120.4
4	1	121.2
5	1	118.7
6	1	119.3
7	1	119.6
8	2	122.1
9	2	123.9
10	2	123.6
11	2	122.4
12	2	122.7
13	3	122.4

Figure 11: Partial view of the four treatments' data.

In this example, there are four treatments (different machine

settings), and the first question of interest is: Are there any differences in the machine settings with regard to the resulting fill weights?

In this particular example, the data are continuous and likely close enough to being normally distributed. Hence, they can be modeled using a normal distribution. The variances of each treatment are also similar, so we can proceed with the ANOM.

The first step is to run the ANOMB.jsl

script. Figure 12 shows the dialog box that appears when you run the script. Now simply enter the columns of interest and select the desired level of significance.

The output, shown in Figure 13, includes the decision limits, a chart, a summary of the group means and standard deviations, as well as the ability to change the level of significance.

From this chart, you can conclude (at




Figure 12: The ANOM script dialog.




Figure 13: ANOM script output for fweights.jmp.

significance level $\alpha = 0.05$) that there are differences in the treatments. In particular, Treatment 1 underfills the tubes, and Treatment 2 overfills the tubes when compared to the overall fill level.

Unbalanced Example: Stream Remediating Data

Figure 14 gives the data of toxin levels in fish collected from three locations along a stream.

Due to lab error, two samples from one site were destroyed, resulting in unbalanced data. The dialog box produced by running the script ANOMUB.jsl on Fish.jmp is the same as the one shown in Figure 12. The output for this example is shown in Figure 15.

One concludes that all three fish samples have toxins levels that are different than the overall mean, with site A and C having low toxin levels and site B having high levels.

	Location	Toxin mg/kg
1	A	3.8
2	A	3.2
3	A	3.5
4	B	5.3
5	B	5.7
6	B	5.6
7	B	5.8
8	B	5.6
9	C	3.3
10	C	2.9
11	C	3.8
12	C	3.3
13	C	3.4

Figure 14: Fish.jmp




Figure 15: ANOM script output for Fish.jmp.

Beyond the Scripts

If you have more than eight means or significance levels other than 0.1, 0.05, 0.01, or 0.001, you must add additional critical values to the look-up files or compute the decision limits and then modify a JMP control chart by removing the control chart limits and then adding reference lines corresponding to the ANOM decision limits.

The ANOM provides a simple method for comparing means that includes a graphical component that clearly indicates when means differ and which means differ. The two scripts described here easily implement ANOM in JMP.

References

Mason, R. L.; Gunst, R. F.; and Hess, J. L. (1989). *Statistical Design and Analysis of Experiments With Applications to Engineering and Science*. John Wiley & Sons, New York, NY.

Nelson L. S. (1983). “Exact Critical Values for Use with the Analysis of Means” *Journal of Quality Technology* 15, pp. 40 -44.

Nelson, P. R. (1982). “Exact Critical Points for the Analysis of Means” *Communications in Statistics – Theory and Methods* 11(6), pp. 699-709.

Nelson, P. R., Coffin, M. and Copeland, K. A. F. (2003). *Introductory Statistics for Engineering Experimentation*. Academic Press, New York, NY.

Nelson, P. R., Wyludyka, P. and Copeland, K. A. F. (2005). *The Analysis of Means: A Graphical Method for Comparing Means, Rates, and Proportions*.

Ott, E. R. (1967). “Analysis of Means – A Graphical Procedure.” *Industrial Quality Control* 24, pp. 101-109. Reprinted in *Journal of Quality Technology* 15 (1983), pp. 10-18.

Vardeman, S. B. (1994). *Statistics for Engineering Problem Solving*. PWS Publishing Company, Boston, MA.

The author wishes to thank Erin Vang and Mark Bailey of JMP for their help in creating the ANOM scripts.

Creating a List of Open Data Tables

Jeff Perkinson, JMP Technical Support

A recent caller to technical support asked how to get a list of all open data tables and present the list in a dialog box.

While there's no command that returns a list of open tables, it's easy to construct the list with a loop using the `N Table()` and `Data Table()` commands. The `N Table()` command returns the number of open data tables, and the `Data Table()` command takes a numeric argument and returns a reference to the n^{th} open table.

The `ListTables.jsl` script shown below, which can be downloaded from <http://www.jmp.com/news/jmpcable>, uses those commands in a `For()` loop to construct a list. The list is then used in a window. This window allows the user to select an open data table from a list box and view a list of the columns in that table, as shown in Figure 16.




Figure 16: The script produces a list of open data tables and their columns.

```
list={};

for(i=1, i<=n table(), i++,
  list[i]=data table(i) << get name;
);

win=new window(
  "Choose a data table",
  hb=hlist box(
    panel box(
      "Choose a data table",
      dt=list box(list, max selected(1))
    ),
    selectbutton=button box(
      "Get Columns",
      pb<<delete;
      current data table(data table((dt<<get selected)[1]));
      hb << append(
        pb=panel box(
          "Columns in data table " || (dt<<get selected)[1],
          col list box(all)
        )
      )
    ),
    pb=panel box(
      "Columns in data table",
      cols = col list box()
    )
  )
);
```

Look for JMP at these Conferences

February 7-8, 2005	American Society for Quality (ASQ) Six Sigma Conference in Palm Springs, CA
February 16-17, 2005	Six Sigma in Chemicals and Pharmaceuticals in Dallas, TX
March 8, 2005	Central Ohio ASQ Conference in Columbus, OH
March 16, 2005	61st Annual ASQ Rochester Section Quality Conference in Rochester, NY
April 7, 2005	25th Annual Boston Quality Conference (BOSCON 2005) in Woburn, MA
April 10-13, 2005	SUGI (SAS Users Group International) in Philadelphia, PA
April 19-21, 2005	Quality Expo 2005 in Rosemont, IL
May 16-18, 2005	World Conference on Quality and Improvement in Seattle, WA
May 18-19, 2005	IMACS 2005 (International Meeting on Automated Compliance Systems) in Princeton, NJ

Meet the Trainer: An Interview

To give you the inside scoop on JMP training instructors, this column features interviews with JMP trainers. It describes their statistics background and interests, along with a few fun facts you can bring up in class.

Introducing....Marissa Langford

This feature spotlights Marissa Langford, a full-time JMP instructor in the Statistical Training & Technical Services group.



Marissa Langford

JMPer Cable: Hi Marissa. What is your background and how did you become interested in JMP training?

Marissa: I received a bachelor's and master's degree in textile engineering and a minor in statistics from North Carolina State University (NCSU). I worked eight years in the textile industry between my undergraduate and graduate programs. One of my statistics professors used JMP in two courses. I liked the software because it was extremely easy to learn and use. Therefore, when I taught an undergraduate quality control course

for textile engineering students after I completed my graduate degree, I used JMP in the class. That was my first experience teaching JMP software.

I came to SAS in May 2000 as a Technical Training Specialist on the SAS programming side of the Education division. I left in February 2002 to stay at home with my first child. Lori Rothenberg, manager of the JMP training group, approached me not long after and invited me to be a JMP contract instructor.

JMPer Cable: What is your role within JMP training?

Marissa: I am currently a Statistical Services Specialist; I teach the data exploration, ANOVA and regression, design and analysis of experiments, statistical quality control, and introduction to the JMP Scripting Language (JSL) courses. I also participate in customer conference calls to determine their needs and to match our training to those needs. I enjoy this because it provides a connection with customers.

Additionally, I helped revise most of the JMP courses and I am currently converting the ANOVA and regression course to the live web format.

I am also starting to work with the Design Institute for Six Sigma. After I complete my black belt training, I will be learning and then delivering the courses offered through the Design Institute for Six Sigma.

JMPer Cable: What have been some of your favorite projects?

Marissa: Last year I worked on the curriculum development team for the SAS Microarray Solution courses. This is the only solution that utilizes both SAS and JMP. SAS is used to fit models and the results are surfaced in JMP. After the user has results in JMP, they can use JMP tools to further explore their data. I worked some on the user course, but I focused primarily on the JMP and Statistical Essentials for SAS Microarray Solution course. I enjoyed this project because I had the opportunity to learn about an entirely new subject.

I have also worked on many JMP course revision properties. These are always fun because you get to put your stamp on a course.

JMPer Cable: What do you like best about training?

Marissa: My favorite thing about training is seeing a student "get it." I actually like teaching the JMP Software: Statistical Data Exploration class because this is typically the first course that newcomers take. I like showing them the possibilities of JMP and talking with them about how they can use JMP in their jobs.

The JMP scripting course is also one of my favorites to teach. Most students in this course are advanced. They are typically proficient in using JMP, and they are eager to learn how to automate tasks. It is a highlight to hear

(continued from page 11)

them discuss how they can use the tools I've just given them.

One of my new responsibilities is to learn the Six Sigma courses. I'm enjoying this because I worked in quality in the textile industry and I based my thesis on innovative quality systems for the textile industry. I believe in improving our quality—and improving our processes—so we make the best product at the best price. I like bringing that message to people and giving them the tools they need.

JMPer Cable: Where do you get your inspiration?

Marissa: I honestly get my inspiration and enthusiasm from the students and I think that is true for any instructor. If the students are enjoying the class, I'm enjoying the class. It doesn't hurt that I think JMP is truly a superior software and I want to get that message to as many people as possible.

JMPer Cable: Any other information you'd like to share?

Marissa: I'm the mom of two great kids (JT will be three in November and Annabelle will be one in January). I'm fortunate that the folks I work for at SAS understand that I have a young family and they encourage me to find a balance between family and work.

About JMPer Cable

Issue 16 Winter 2005

JMPer Cable is mailed to JMP users who are registered users with SAS Institute. It is also available online at www.jmp.com.

Contributors

Karen Copeland
Matt Flynn
Marissa Langford
Wayne Levin
Jeff Perkinson

Technical Editor

Ann Lehman

Editor and Designer

Meredith Blackwelder

Printing

SAS Institute Print Center

Questions or comments

jmp@sas.com

To order JMP software

1-877-594-6567

For more information on JMP

1-877-594-6567

www.jmp.com

Copyright 2004, SAS Institute. All rights reserved. SAS, JMP, JMPer Cable, and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates registration. Other brand and product names are trademarks of their respective companies.